
1. [bookmark: _Toc122434485][bookmark: _Toc65842567]Overview
This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 11.1.4 which is part of the NR5GC IRAT test suite.
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.
Contact:	Kalahasti, Narendra
	Narendra.Kalahasti@anritsu.com
0. [bookmark: _Toc51622099][bookmark: _Toc55479332][bookmark: _Toc60751530][bookmark: _Toc60757075][bookmark: _Toc65842568]Verification Test Summary 
Test Case: 	11.1.4
ATS Version:	iwd-TTCN3-B2020-09_D20wk50
[bookmark: _Hlk37919671]System Simulator used:	Anritsu Protocol Conformance Test System ME7834NR
UE used:	MediaTek MT6297
Verification Status:	PASS


1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc65842569]Corrections required
1.1 [bookmark: _Toc54186269][bookmark: _Toc65842570][bookmark: OLE_LINK12][bookmark: OLE_LINK13][bookmark: OLE_LINK2][bookmark: OLE_LINK3][bookmark: OLE_LINK10][bookmark: OLE_LINK11]Change 1
	Function name
	fl_TC_11_1_4_NR5GC_EUTRA_TestBody

	Reason for change
	1. AS given in 38.523-1 Table 11.1.4.3.2-1 after step 17: The following messages are to be observed on E-UTRA Cell 1 unless explicitly stated otherwise.
Several occurrences of eutra_cell2 needs to be replaced by eutra_cell1. Except in case of SysInfo coordination message the eutra_cell2 needs to be replaced with eutra_cell3 based on change in R5-206360 that this test is using an inter-frequency pair of cells (eutra_cell1 and 3) instead of intra-frequency pair (eutra_cell1 and 2).
2. ActiveFlag can have any value in TAU request in step 18 based on R5s210288 and associated prose CR.
3. Based on 24.301 cl 8.2.29.8 -  lastVisitedRegisteredTai IE shall be included if the UE holds a valid last visited registered. 
UE holds it from the test preamble.
4. Between steps 18 and 19, initial security activation as well as establishment of data radio bearer associated with default EPS bearer are missing. That would correspond to steps 5-8 from 36.580 cl 4.5A.6.3-1.

Prose CR is associated with this change

5. IP handling on secured ports for EUTRA cell have been removed from DRBMUX during de-REGESITER in LTE, they remain only for NR cell. So it is necessary to add them explicitly into DRBMUX routing table before redirection.
6. Based on 34.229-1 C.32 steps 4-5, EPS bearer deactivation procedure for UE with E-UTRA support are missing.
7. Based on R5s210288 and associated prose CR the ping test should be removed for this IMS test case.

	Summary of change
	1. Several occurrences of eutra_cell2 be replaced by eutra_cell1. Except in case of SysInfo coordination message the eutra_cell2 replaced with eutra_cell3
2. Template for Active flag in TAU request in step 18 matches both values.
3. Default 3rd parameter called by f_EUTRA38_TrackingAreaUpdateFromN1 is used instead of “omit”.
4. Added procedures corresponding to steps 5-8 from 36.580 cl 4.5A.6.3-1.
5. Explicit addition of secured ports on EUTRA cell of DRB1 in DRBMUX before the redirection.
6. At the end of test body added procedures corresponding to EPS bearer deactivation.
7. Removed ping test.

	TTCN module
	EPS_Fallback_NR5GC_EUTRA.ttcn

	MCC160 Comment
	



Before Change:
	  function fl_TC_11_1_4_NR5GC_EUTRA_TestBody() runs on EUTRA_Multi5GS_PTC
  {
    var template (value) EUTRA38_DRBInfoList_Type v_DRBIdList := f_EUTRA38_MobileInfo_GetDRBInfoList();
    var integer v_DRBId := valueof(v_DRBIdList[0].DRB_Id);
    var template (value) RadioBearerId_Type v_RbId_DRB := cs_RadioBearerId_DRB(v_DRBId);
    
    //Eutra Cell1 is set as barred and modified Sysinfo updated, setting the Cell power to suitable neighbour cell FFS : Prose CR ??
    f_EUTRA_CellInfo_SetSysInfo_CellBarred(eutra_Cell3, barred); // @sic R5-206360 sic@
    f_EUTRA_ModifySysinfo(eutra_Cell3); // @sic R5-206360 sic@

    // @siclog "Step 1" siclog@
    f_EUTRA38_SetCellPower(eutra_Cell1, tsc_SuitableCellRS_EPRE, tsc_SuitableCellRS_EPRE); // @sic R5-206360 sic@
    f_EUTRA38_SetCellPower(eutra_Cell3, tsc_ServingCellRS_EPRE, tsc_ServingCellRS_EPRE); // @sic R5-206360 sic@
    
    // Send EUTRA frequency to NR and wait for NR info
    f_EUTRA_NR_SendSysinfoCoOrd(NR, f_EUTRA38_IRAT_ListOfFrequencies({eutra_Cell1, eutra_Cell2}));
    
    // @siclog "Step 18" siclog@
    //Generic test procedure in TS 38.508-1 [4] Table 4.9.7.2.2-1 is performed without 'connected without release' on E-UTRAN cell upon mobility from NR.
    f_EUTRA38_TrackingAreaUpdateFromN1(eutra_Cell2, tsc_RRC_TI_Def, omit, -, noRrcConnectionRelease);


    // Tell IMS PTC the RAT has changed
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1], cms_IPCAN_IMS_QueryResponse(f_EUTRA_IMS_IpcanInfo()));

    // @siclog "Step 19 - 21" siclog@
    f_EUTRA_ActivateDedicatedEpsBearer_Speech(eutra_Cell2);

    // wait for Trigger for Step 22
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);

    // @siclog "Step 22" siclog@
    f_Delay(1.0);

    // @siclog "Step 23a1" siclog@
    if (pc_IP_Ping) {
      f_EUTRA_CheckDataPath(eutra_Cell2, PDCP, defaultBearer, "Step 23a", v_RbId_DRB);  // @sic R5-204399 sic@
    }

    //Now send the Coord to IMS to release the call
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]);
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    
    //Send a trigger to NR for it to release its cell
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("Postamble"));

  }



[bookmark: OLE_LINK5][bookmark: OLE_LINK4]
After Change:
	  
    function fl_TC_11_1_4_NR5GC_EUTRA_TestBody() runs on EUTRA_Multi5GS_PTC
  {
    var template (value) EUTRA38_DRBInfoList_Type v_DRBIdList := f_EUTRA38_MobileInfo_GetDRBInfoList();
    var integer v_DRBId := valueof(v_DRBIdList[0].DRB_Id);
    var template (value) RadioBearerId_Type v_RbId_DRB := cs_RadioBearerId_DRB(v_DRBId);
    var NasCount_Type v_NasCount;
    var PDN_AddressInfo_Type v_PDN_AddressInfo;
    var template (value) IP_AddrInfo_Type v_IPv4Addr, v_IPv6Addr;
 
    //Eutra Cell1 is set as barred and modified Sysinfo updated, setting the Cell power to suitable neighbour cell FFS : Prose CR ??
    f_EUTRA_CellInfo_SetSysInfo_CellBarred(eutra_Cell3, barred); // @sic R5-206360 sic@
    f_EUTRA_ModifySysinfo(eutra_Cell3); // @sic R5-206360 sic@

    // @siclog "Step 1" siclog@
    f_EUTRA38_SetCellPower(eutra_Cell1, tsc_SuitableCellRS_EPRE, tsc_SuitableCellRS_EPRE); // @sic R5-206360 sic@
    f_EUTRA38_SetCellPower(eutra_Cell3, tsc_ServingCellRS_EPRE, tsc_ServingCellRS_EPRE); // @sic R5-206360 sic@
    
    // Send EUTRA frequency to NR and wait for NR info
    f_EUTRA_NR_SendSysinfoCoOrd(NR, f_EUTRA38_IRAT_ListOfFrequencies({eutra_Cell1, eutra_Cell3}));
    
    // @siclog "Step 18" siclog@
    //Generic test procedure in TS 38.508-1 [4] Table 4.9.7.2.2-1 is performed without 'connected without release' on E-UTRAN cell upon mobility from NR.
    
        
    v_NasCount := f_EUTRA38_TrackingAreaUpdateFromN1(eutra_Cell1, tsc_RRC_TI_Def, -, -, noRrcConnectionRelease,-,-,-,-,-,(tsc_EpsUpdate_NotActive,tsc_EpsUpdate_Active));

    // Steps 5-8 of 4.5A.6.3-1
    f_EUTRA38_MultiPDN_Activate_SRB2_DRB_SendRrcMsg(eutra_Cell1,
                                         tsc_RRC_TI_Def,
                                         v_NasCount,-,-,-,-,true);
    
    f_EUTRA_Activate_SRB2_DRB_ReceiveRrcAndNasMsg(eutra_Cell1,
                                                  tsc_RRC_TI_Def);

    
    v_PDN_AddressInfo := f_PDN_AddressInfo_Get(PDN_1);
    v_IPv4Addr := cs_IPv4Addr(v_PDN_AddressInfo.PCSCF_IPAddressIPv4);
    v_IPv6Addr := cs_IPv6Addr(v_PDN_AddressInfo.PCSCF_IPAddressIPv6); 

    f_IP_AddExplicitRouting_TCP_UDP(IP, 5061, v_IPv4Addr, cs_DrbInfo_EUTRA(eutra_Cell1, tsc_DRB1));
    f_IP_AddExplicitRouting_TCP_UDP(IP, 5061, v_IPv6Addr, cs_DrbInfo_EUTRA(eutra_Cell1, tsc_DRB1));
    f_IP_AddExplicitRouting_TCP_UDP(IP, 5062, v_IPv4Addr, cs_DrbInfo_EUTRA(eutra_Cell1, tsc_DRB1));
    f_IP_AddExplicitRouting_TCP_UDP(IP, 5062, v_IPv6Addr, cs_DrbInfo_EUTRA(eutra_Cell1, tsc_DRB1));

    // Tell IMS PTC the RAT has changed
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1], cms_IPCAN_IMS_QueryResponse(f_EUTRA_IMS_IpcanInfo()));

    // @siclog "Step 19 - 21" siclog@
    f_EUTRA_ActivateDedicatedEpsBearer_Speech(eutra_Cell1);

    // wait for Trigger for Step 22
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);

    // @siclog "Step 22" siclog@
    f_Delay(1.0);

    // @siclog "Step 23a1" siclog@
//REMOVED /*    if (pc_IP_Ping) {
      f_EUTRA_CheckDataPath(eutra_Cell1, PDCP, defaultBearer, "Step 23a", v_RbId_DRB);  // @sic R5-204399 sic@
    } */

    //Now send the Coord to IMS to release the call
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]);
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    
    
    f_EUTRA_DeactivateEPS_BearerContext (eutra_Cell1, tsc_EpsDedicatedBearerId);

    //Send a trigger to NR for it to release its cell
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("Postamble"));

  }




1.2 [bookmark: _Toc62594273][bookmark: _Toc65842571]Change 2
	Function name
	fl_TC_11_1_4_NR5GC_TestBody

	Reason for change
	1. No need to send IRAT sysinfo in function fl_TC_11_1_3_NR5GC_TestBody
2. Based on 38.508 4.9.16 the establishmentCause should be mtAccess and ServiceType should be MobileTerminatedServices.
3. Because of R5-206360, removing FFS comment and correcting the fact that reselection priority 1 is assigned to eutra_cell3 (38.523-1 table 11.1.4.3.3-6).
4. The same change as in R5s210149 change 3 bullet 1 is needed in step 17 – the timer inside f_NR_RRCRelease_Common needs to be disabled.

	Summary of change
	1. Removed f_EUTRA_NR_SendSysinfoCoOrd at the beginning of test body.
2. Changed template matching for establishmentCause and ServiceType in steps 3-9
3. Removed FFS comment and assigned reselection priority 1 to eutra_cell3
4. Signal to disable t_Timer in the RRC release procedure.

	TTCN module
	11_1\EPS_Fallback_NR5GC_NR.ttcn

	MCC160 Comment
	



Before Change:
	    function fl_TC_11_1_4_NR5GC_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var integer v_DRBId;
    var template GSM_MobilityInfo_Type v_PDUSessionInfo;
    var template (value) RedirectedCarrierInfo v_RedirectedCarrierInfo;
    var EUTRA_NR_CoOrd_SysInfo_Type v_CoOrd_SysInfo := f_EUTRA_NR_WaitForCoOrd_SysInfo(EUTRA);

    // return the Sysinfo before the test body really starts
    f_EUTRA_NR_SendSysinfoCoOrd(EUTRA, f_NR5GC_IRAT_ListOfFrequencies());

    
    //@siclog "Step 2" siclog@
    f_NR_Paging_Def(nr_Cell1, RRC_IDLE);
    
    //@siclog "Steps 3 - 9" siclog@
    v_ReceivedMsg := f_NR5GC_RRCIdleToConnected_Steps2_4(nr_Cell1, '0001'B, mo_Data); // @sic R5-206360 sic@


    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell1, v_ReceivedMsg);

    //@siclog "Step 10a1" siclog@
    //Check: Does the test result of generic test procedure in TS 38.508-1[4] subclause 4.9.1 indicate that the UE is capable of exchanging IP data on DRB#2 associated with the first PDU session?
    if (pc_IP_Ping) {
        v_PDUSessionInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), Internet_DNN);
         if (isvalue(v_PDUSessionInfo)) {
          v_DRBId := f_NR_GetDefaultDRB_InDRBList (f_NR_GetDRBList_ForGivenPDUSession(oct2int(valueof(v_PDUSessionInfo.SessionId))));
          f_NR5GC_CheckDataPath(nr_Cell1, f_NR5GC_GetPdnIndex(valueof(v_PDUSessionInfo)), dedicatedBearer, v_DRBId, "Step 10a1");
         } else {
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "No Internet PDU Session");
        }
    }

    //@siclog "Step 11" siclog@
    f_UT_RequestIMSCall(UT, px_IMS_CalleeUri);
    //@siclog "Step 12 - 16" siclog@
    //Steps 1-5 of the MTSI MO speech call for 5GS procedure according TS 34.229-1 [35] subclause C.21g take place, done on IMS PTC
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);

    //@siclog "Step 17" siclog@
    //The SS transmits an RRCRelease message (IE redirectedCarrierInfo including EUTRA cells).
    //FFS : Why do the frequency info of both Eutra cells needs to be provided when they are both on same frequency ?? Prose CR needed ?

    v_RedirectedCarrierInfo := cs_NR_RedirectedCarrierInfoToEUTRA(v_CoOrd_SysInfo.Eutra[0].Arfcn);

    f_NR_RRCRelease_Common(nr_Cell1, cs_38508_RRCRelease(tsc_NR_RRC_TI_Def,
                                                         -,
                                                         v_RedirectedCarrierInfo,
                                                         cs_CellReselectionPriorities ({cs_FreqPriorityEUTRA_NRDef(v_CoOrd_SysInfo.Eutra[0].Arfcn, 1),
                                                                                        cs_FreqPriorityEUTRA_NRDef(v_CoOrd_SysInfo.Eutra[0].Arfcn, 0)}
                                                                                       )));
    
    // Send both authentication and NR sysinfo paramters to EUTRA
    f_NR5GC_TrackingAreaUpdateFromN1(nr_Cell1);

    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA, "Postamble");

  }
    



After Change:
	    function fl_TC_11_1_4_NR5GC_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var integer v_DRBId;
    var template GSM_MobilityInfo_Type v_PDUSessionInfo;
    var template (value) RedirectedCarrierInfo v_RedirectedCarrierInfo;
    var EUTRA_NR_CoOrd_SysInfo_Type v_CoOrd_SysInfo := f_EUTRA_NR_WaitForCoOrd_SysInfo(EUTRA);

    // return the Sysinfo before the test body really starts
    // REMOVED f_EUTRA_NR_SendSysinfoCoOrd(EUTRA, f_NR5GC_IRAT_ListOfFrequencies());
    
    //@siclog "Step 2" siclog@
    f_NR_Paging_Def(nr_Cell1, RRC_IDLE);
    
    //@siclog "Steps 3 - 9" siclog@
    v_ReceivedMsg := f_NR5GC_RRCIdleToConnected_Steps2_4(nr_Cell1, '0010'B, mt_Access); // @sic R5-206360 sic@
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell1, v_ReceivedMsg);

    //@siclog "Step 10a1" siclog@
    //Check: Does the test result of generic test procedure in TS 38.508-1[4] subclause 4.9.1 indicate that the UE is capable of exchanging IP data on DRB#2 associated with the first PDU session?
    if (pc_IP_Ping) {
        v_PDUSessionInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), Internet_DNN);
         if (isvalue(v_PDUSessionInfo)) {
          v_DRBId := f_NR_GetDefaultDRB_InDRBList (f_NR_GetDRBList_ForGivenPDUSession(oct2int(valueof(v_PDUSessionInfo.SessionId))));
          f_NR5GC_CheckDataPath(nr_Cell1, f_NR5GC_GetPdnIndex(valueof(v_PDUSessionInfo)), dedicatedBearer, v_DRBId, "Step 10a1");
         } else {
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "No Internet PDU Session");
        }
    }

    //@siclog "Step 11" siclog@
    f_UT_RequestIMSCall(UT, px_IMS_CalleeUri);
    //@siclog "Step 12 - 16" siclog@
    //Steps 1-5 of the MTSI MO speech call for 5GS procedure according TS 34.229-1 [35] subclause C.21g take place, done on IMS PTC
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);

    //@siclog "Step 17" siclog@
    //The SS transmits an RRCRelease message (IE redirectedCarrierInfo including EUTRA cells).
    v_RedirectedCarrierInfo := cs_NR_RedirectedCarrierInfoToEUTRA(v_CoOrd_SysInfo.Eutra[0].Arfcn);
    //There is timer inside f_NR_RRCRelease_Common which needs to expire before the NR5GC component coordinate with EUTRA, however if UE starts new connection on EUTRA before that it fails the test. The same change was in 11.1.1
    f_NR_RRCRelease_Common(nr_Cell1, cs_38508_RRCRelease(tsc_NR_RRC_TI_Def,
                                                         -,
                                                         v_RedirectedCarrierInfo,
                                                         cs_CellReselectionPriorities ({cs_FreqPriorityEUTRA_NRDef(v_CoOrd_SysInfo.Eutra[1].Arfcn, 1), //Priority 1 has the EARFCN of cell3 (Table 11.1.4.3.3-6:)
                                                                                        cs_FreqPriorityEUTRA_NRDef(v_CoOrd_SysInfo.Eutra[0].Arfcn, 0)}
                                                                                       )),-,true);
    
    // Send both authentication and NR sysinfo paramters to EUTRA
    f_NR5GC_TrackingAreaUpdateFromN1(nr_Cell1);

    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA, "Postamble");

  }




1. [bookmark: _Toc58491298][bookmark: _Toc60751532][bookmark: _Toc60757078][bookmark: _Toc65842572]Branches Executed
The test case was executed on NR band n78.
1. [bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666][bookmark: _Toc51622108][bookmark: _Toc56072723][bookmark: _Toc65842573]Execution Log Files 
3.1 [bookmark: _Toc65842574]MediaTek MT6297
The MediaTek MT6297 UE passed this test case on Anritsu Protocol Conformance Test System ME7834NR. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
TC_11_1_4_LOG.xml
· PICS/PIXIT parameter file:
TC_11_1_4_PIXIT.txt



4 [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc51622110][bookmark: _Toc56072725][bookmark: _Toc65842575]References
	[1]
	R5s210300



